In Class Demonstration: The Change of Coordinate Matrix

By Russell Blyth

```
> with(LinearAlgebra):
```

Let V be a vector space of finite dimension, with bases B and B'. The change of coordinate matrix Q has columns which are the coordinate vectors for the vectors in B' relative to B.

We look at an example in R^2. The matrices B and Bp contain the respective basis vectors as columns.

```
> B := <<-2,3>|<2,-1>>;
```

$$B := \begin{bmatrix} -2 & 2 \\ 3 & -1 \end{bmatrix} \tag{1}$$

```
> Bp := <<2,-2>|<1,5>>;
```

$$Bp := \begin{bmatrix} 2 & 1 \\ -2 & 5 \end{bmatrix} \tag{2}$$

Find the coordinates for the Bp vectors relative to B.

```
> Q1 := LinearSolve(B,Column(Bp,1));
```

$$Q1 := \begin{bmatrix} -\frac{1}{2} \\ \frac{1}{2} \end{bmatrix} \tag{3}$$

```
> Q2 := LinearSolve(B,Column(Bp,2));
```

$$Q2 := \begin{bmatrix} \frac{11}{4} \\ \frac{13}{4} \end{bmatrix} \tag{4}$$

```
> Q := <Q1|Q2>;
```

$$Q := \begin{bmatrix} -\frac{1}{2} & \frac{11}{4} \\ \frac{1}{2} & \frac{13}{4} \end{bmatrix} \tag{5}$$

This matrix Q changes coordinates from B' to B.

Try it out. Suppose x is a vector whose coordinates relative to B' we know.

```
> xBp := <-2,-3>;
```

$$xBp := \begin{bmatrix} -2 \\ -3 \end{bmatrix} \tag{6}$$

Now find the coordinates for x relative to B, using Q.

```
> xB := Q.xBp;
```

(7)

$$xB := \begin{bmatrix} \frac{-29}{4} \\ \frac{-43}{4} \end{bmatrix} \quad (7)$$

Is this right? Let's check;

```
> x1 := xBp[1]*Column(Bp,1)+xBp[2]*Column(Bp,2);
```

$$x1 := \begin{bmatrix} -7 \\ -11 \end{bmatrix} \quad (8)$$

```
> x2 := xB[1]*Column(B,1)+xB[2]*Column(B,2);
```

$$x2 := \begin{bmatrix} -7 \\ -11 \end{bmatrix} \quad (9)$$

We see that the result is the same vector (now expressed in terms of the standard basis)

Now let's try out a linear operator on R^3. Suppose T is the linear operator such that T(1,2,3) = (4,5,6), T (2,2,2) = (-1,-1,-1), and T(1,0,1) = (0,1,0). We will find the formula for T (in terms of the standard basis).

Let B be the standard basis for R^3, and let B' be the basis {(1,2,3),(2,2,2), (1,0,1)}.

We write these bases as columns of the matrices B3 and B3p:

```
> B3 := <<1,0,0>|<0,1,0>|<0,0,1>>;
```

$$B3 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad (10)$$

```
> B3p := <<1,2,3>|<2,2,2>|<1,0,1>>;
```

$$B3p := \begin{bmatrix} 1 & 2 & 1 \\ 2 & 2 & 0 \\ 3 & 2 & 1 \end{bmatrix} \quad (11)$$

Find the matrix of T relative to B'. To do this we need to find the coordinates of the image of each basis vector relative to B'; these coordinate vectors then become the columns of the matrix of T relative to B'.

```
> TIm := <<4,5,6>|<-1,-1,-1>|<0,1,0>>;
  T1 := LinearSolve(B3p,Column(TIm,1));
  T2 := LinearSolve(B3p,Column(TIm,2));
  T3 := LinearSolve(B3p,Column(TIm,3));
```

$$TIm := \begin{bmatrix} 4 & -1 & 0 \\ 5 & -1 & 1 \\ 6 & -1 & 0 \end{bmatrix}$$

$$T1 := \begin{bmatrix} 1 \\ \frac{3}{2} \\ 0 \end{bmatrix}$$

$$T2 := \begin{bmatrix} 0 \\ \frac{-1}{2} \\ 0 \end{bmatrix}$$

$$T3 := \begin{bmatrix} 0 \\ \frac{1}{2} \\ -1 \end{bmatrix} \quad (12)$$

```
> TB3p := <T1|T2|T3>;
```

$$TB3p := \begin{bmatrix} 1 & 0 & 0 \\ \frac{3}{2} & \frac{-1}{2} & \frac{1}{2} \\ 0 & 0 & -1 \end{bmatrix} \quad (13)$$

Now find the matrix Q which gives the change of coordinates. In this case it is easy to write the change of coordinate matrix from B' to B - the columns are just the vectors in B'.

```
> Q3 := B3p;
```

$$Q3 := \begin{bmatrix} 1 & 2 & 1 \\ 2 & 2 & 0 \\ 3 & 2 & 1 \end{bmatrix} \quad (14)$$

Find the inverse of Q. This gives the change of coordinate matrix from B to B'.

```
> Q3inv := Q3^(-1);
```

$$Q3inv := \begin{bmatrix} \frac{-1}{2} & 0 & \frac{1}{2} \\ \frac{1}{2} & \frac{1}{2} & \frac{-1}{2} \\ \frac{1}{2} & -1 & \frac{1}{2} \end{bmatrix} \quad (15)$$

The matrix of T relative to the standard basis is now computed as
Q * (T rel to b') * Qinv

```
> TB3 := Q3.TB3p.Q3inv;
```

(16)

$$TB3 := \begin{bmatrix} \frac{-5}{2} & \frac{-1}{2} & \frac{5}{2} \\ \frac{-5}{2} & \frac{-3}{2} & \frac{7}{2} \\ \frac{-7}{2} & \frac{-1}{2} & \frac{7}{2} \end{bmatrix} \quad (16)$$

Compute for a general vector now

```
> formula := TB3.<a1,a2,a3>;
```

$$formula := \begin{bmatrix} -\frac{5}{2}\,a1 - \frac{1}{2}\,a2 + \frac{5}{2}\,a3 \\ -\frac{5}{2}\,a1 - \frac{3}{2}\,a2 + \frac{7}{2}\,a3 \\ -\frac{7}{2}\,a1 - \frac{1}{2}\,a2 + \frac{7}{2}\,a3 \end{bmatrix} \quad (17)$$

Check that the vectors of the basis B' are mapped correctly - the coordinates are relative to the standard basis.

```
> TB3.Column(B3p,1);
```

$$\begin{bmatrix} 4 \\ 5 \\ 6 \end{bmatrix} \quad (18)$$

```
> TB3.Column(B3p,2);
```

$$\begin{bmatrix} -1 \\ -1 \\ -1 \end{bmatrix} \quad (19)$$

```
> TB3.Column(B3p,3);
```

$$\begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix} \quad (20)$$

```
>
```